// EverSequence GE G35 light control protocol converter for EverSequence

 // By Andrew Coulson, aka AustinLightGuy <acoulson@practical-apps.com>

 // sequencer program (http://practical-apps.com/practical-apps/products/utilities/eversequence.aspx)

 // 

 // Protocol to talk to this driver is:

 // ".LIRGB", where "." is ascii period character, L is a byte representing the light number to 

 // address (between 1 and XMAS_LIGHT_COUNT), I is a byte representing intensity (usually 0xCC),

 // and RGB are bytes representing red, green and blue intensity levels (between 0x00 and 0x0F)

 // Based on this code:  

 /*!    

 **    GE Christmas light control for Arduino  

 **    Ported by Scott Harris <scottrharris@gmail.com>  

 **    scottrharris.blogspot.com  

 **    http://scottrharris.blogspot.com/2010/12/controlling-ge-color-effects-lights.html

 **

 **    which is in turn base on:

 **

 **     Christmas Light Control  

 **     By Robert Quattlebaum <darco@deepdarc.com>  

 **     Released November 27th, 2010  

 **  

 **     For more information,  

 **     see <http://www.deepdarc.com/2010/11/27/hacking-christmas-lights/>.

 **  

 **     Originally intended for the ATTiny13, but should  

 **     be easily portable to other microcontrollers.  

 */  

 #define xmas_color_t uint16_t // typedefs can cause trouble in the Arduino environment  

 // Eliminate the .h file  

 #define XMAS_LIGHT_COUNT          (5) // originally 50 or 36, but I want to be able to control many alternating lights 

                                       // on the string via one "channel" from my sequencer, so I use a divisor

                                       // of 50 so that, for example, every 5th light, starting at position 1,

                                       // is addressed by address "1".

 #define XMAS_CHANNEL_MAX          (0xF)  

 #define XMAS_DEFAULT_INTENSITY     (0xCC)  

 #define XMAS_HUE_MAX               ((XMAS_CHANNEL_MAX+1)*6-1)  

 #define XMAS_COLOR(r,g,b)     ((r)+((g)<<4)+((b)<<8))  

 #define XMAS_COLOR_WHITE     XMAS_COLOR(XMAS_CHANNEL_MAX,XMAS_CHANNEL_MAX,XMAS_CHANNEL_MAX)  

 #define XMAS_COLOR_BLACK     XMAS_COLOR(0,0,0)  

 #define XMAS_COLOR_RED          XMAS_COLOR(XMAS_CHANNEL_MAX,0,0)  

 #define XMAS_COLOR_GREEN     XMAS_COLOR(0,XMAS_CHANNEL_MAX,0)  

 #define XMAS_COLOR_BLUE          XMAS_COLOR(0,0,XMAS_CHANNEL_MAX)  

 #define XMAS_COLOR_CYAN          XMAS_COLOR(0,XMAS_CHANNEL_MAX,XMAS_CHANNEL_MAX)  

 #define XMAS_COLOR_MAGENTA     XMAS_COLOR(XMAS_CHANNEL_MAX,0,XMAS_CHANNEL_MAX)  

 #define XMAS_COLOR_YELLOW     XMAS_COLOR(XMAS_CHANNEL_MAX,XMAS_CHANNEL_MAX,0)  

 // Pin setup  

 #define XMASPIN 4 // I drive the LED strand from pin #4  

 #define STATUSPIN 13 // The LED  

 byte inByte=0;              // incoming byte from serial RX

 char inString[6];            // string for incoming serial data

 int cmdPos = 0;

 int lightChannel;

 byte intensity;

 byte redVal;

 byte greenVal;

 byte blueVal;

 // The delays in the begin, one, and zero functions look funny, but they give the correct  

 // pulse durations when checked with a logic analyzer. Tested on an Arduino Uno.  

 void xmas_begin()  

 {  

  digitalWrite(XMASPIN,1);  

  delayMicroseconds(7); //The pulse should be 10 uS long, but I had to hand tune the delays. They work for me  

  digitalWrite(XMASPIN,0);   

 }  

 void xmas_one()  

 {  

  digitalWrite(XMASPIN,0);  

  delayMicroseconds(11); //This results in a 20 uS long low  

  digitalWrite(XMASPIN,1);  

  delayMicroseconds(7);   

  digitalWrite(XMASPIN,0);  

 }  

 void xmas_zero()  

 {  

  digitalWrite(XMASPIN,0);  

  delayMicroseconds(2);   

  digitalWrite(XMASPIN,1);  

  delayMicroseconds(20-3);   

  digitalWrite(XMASPIN,0);  

 }  

 void xmas_end()  

 {  

  digitalWrite(XMASPIN,0);  

  delayMicroseconds(20); // Can be made shorter  

 }  

 // The rest of Robert's code is basically unchanged  

 void xmas_fill_color(uint8_t begin,uint8_t count,uint8_t intensity,xmas_color_t color)  

 {  

      while(count--)  

      {  

           xmas_set_color(begin++,intensity,color);  

      }  

 }  

 void xmas_fill_color_same(uint8_t begin,uint8_t count,uint8_t intensity,xmas_color_t color)  

 {  

      while(count--)  

      {  

           xmas_set_color(0,intensity,color);  

      }  

 }  

 void xmas_set_color(uint8_t led,uint8_t intensity,xmas_color_t color) {  

      uint8_t i;  

      xmas_begin();  

      for(i=6;i;i--,(led<<=1))  

           if(led&(1<<5))  

                xmas_one();  

           else  

                xmas_zero();  

      for(i=8;i;i--,(intensity<<=1))  

           if(intensity&(1<<7))  

                xmas_one();  

           else  

                xmas_zero();  

      for(i=12;i;i--,(color<<=1))  

           if(color&(1<<11))  

                xmas_one();  

           else  

                xmas_zero();  

      xmas_end();  

 }  

 xmas_color_t  

 xmas_color(uint8_t r,uint8_t g,uint8_t b) {  

      return XMAS_COLOR(r,g,b);  

 }  

 xmas_color_t  

 xmas_color_hue(uint8_t h) {  

      switch(h>>4) {  

           case 0:     h-=0; return xmas_color(h,XMAS_CHANNEL_MAX,0);  

           case 1:     h-=16; return xmas_color(XMAS_CHANNEL_MAX,(XMAS_CHANNEL_MAX-h),0);  

           case 2:     h-=32; return xmas_color(XMAS_CHANNEL_MAX,0,h);  

           case 3:     h-=48; return xmas_color((XMAS_CHANNEL_MAX-h),0,XMAS_CHANNEL_MAX);  

           case 4:     h-=64; return xmas_color(0,h,XMAS_CHANNEL_MAX);  

           case 5:     h-=80; return xmas_color(0,XMAS_CHANNEL_MAX,(XMAS_CHANNEL_MAX-h));  

      }  

 }  

 // Read from serial port and call function set intensity and RGB value out via

 // G35 protocol

 //

 // Protocol to talk to this driver is:

 // ".LIRGB", where "." is ascii period character, L is a byte representing the light number to 

 // address (between 1 and XMAS_LIGHT_COUNT), I is a byte representing intensity (usually 0xCC),

 // and RGB are bytes representing red, green and blue intensity levels (between 0x00 and 0x0F)

 void handleSerial() {

  inByte = Serial.read();

  //delay(5);

  if (inByte == '.') {

     cmdPos = 1;

  } else {

    switch (cmdPos) {

      case 0:

        if (inByte == '.') {

          // should be command char

          cmdPos++;

        } else {

            Serial.write("?");

            Serial.write('0'+cmdPos);

          cmdPos = 0; // invalid char, reset cmdPos

        }

        break;

      case 1:

        // should be X

        if ((inByte >= 0) && (inByte <= XMAS_LIGHT_COUNT)) {

          lightChannel = inByte;

          cmdPos++;

        } else {

            Serial.write("?");

            Serial.write('0'+cmdPos);

          cmdPos = 0; // reset cmdPos

        }

        break;

      case 2:

        intensity = inByte;

        cmdPos++;

        break;

      case 3:

        redVal = inByte;

        cmdPos++;

        break;

      case 4:

        greenVal = inByte;

        cmdPos++;

        break;

      case 5:

        blueVal = inByte;

        Serial.write("OK");

        cmdPos = 0;

        xmas_set_color(lightChannel,intensity,XMAS_COLOR(redVal,greenVal,blueVal));

        break;

      default: 

        Serial.write("?D");

        cmdPos = 0;  // invalid char, reset cmdPos

    }

  }

 }

 void setup()  

 {  

  // configure serial communications:

  Serial.begin(115200);      

  delay(3000);  

  pinMode(XMASPIN, OUTPUT);  

  pinMode(STATUSPIN, OUTPUT);  

  uint8_t i;  

  i = 0;

  while( i < 10) {

    xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR_BLACK); //Enumerate all the lights  

    i++;

  }

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR_BLUE); //Make them all blue  

  digitalWrite(STATUSPIN,1);

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR_RED);

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR(14,0,0));

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR(12,0,0));

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR(10,0,0));

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR(8,0,0));

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR(6,0,0));

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR(4,0,0));

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR(2,0,0));

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR(0,0,0));

  delay(50);  

  digitalWrite(STATUSPIN,0);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,XMAS_DEFAULT_INTENSITY,XMAS_COLOR_GREEN);  

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,0xB0,XMAS_COLOR_GREEN);  

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,0x90,XMAS_COLOR_GREEN);  

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,0x70,XMAS_COLOR_GREEN);  

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,0x50,XMAS_COLOR_GREEN);  

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,0x30,XMAS_COLOR_GREEN);  

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,0x10,XMAS_COLOR_GREEN);  

  delay(50);  

  xmas_fill_color(0,XMAS_LIGHT_COUNT,0x00,XMAS_COLOR_GREEN);  

  delay(50);  

  delay(100); 

 }  

 void loop() {  

  if (Serial.available() > 0) {

      handleSerial();

  }

 }   

